

視窗程式請問網站

簡介：

視窗程式(Window)與網頁(Web)程式好像是兩回事，前者執行速度快，後者可以自由上網，各有利弊。目前很多的櫃台管理程式為了操作效率的考量都還是使用視窗程式，但是很多資料目前卻喜歡放在網站裡面。譬如我們要開一家連鎖加盟店，一定會有網站讓顧客瀏覽很多資訊，包括營業時間或價目表等等，各個分店的櫃台程式也必須隨時與這些資訊同步，譬如網站說大減價了！顧客結帳時，櫃台程式如果說還不知道，就會被罵翻了！

但是當分店經理的我，就因此必須隨時上網監視總店資訊，手動更新價目表嗎？那未免也太忙了吧？本單元就介紹一個簡單的程式機制，可以讓視窗程式隨時與網站溝通，商品價格可以直接查詢網站上的資訊。換言之，視窗程式還是可以用網頁程式的 http 通訊機制與網頁交談的！學會這一招以後，就不必每次都很焦慮新系統是要用 Web 或 Windows 模式開發了！即使兩種程式同時進行開發也不會有溝通的問題。

建立會回答價目的網頁

請先啟動 Visual Web Developer 或 Visual Studio 軟體建立一個空的 ASP.NET 網站，從功能表的『網站』→『加入新項目』加入一個文字檔，名稱改為 Price.txt，寫入內容如下：

```
牛肉麵,80
陽春麵,30
魚丸湯,25
肉燥飯,20
```

接著再加入一個新的 Web Form，就是副檔名為 aspx 的網頁，並將它改名為 Price，雙擊頁面空白處，產生 Page_Load 事件框架後寫程式如下：

```
Imports System.IO
Partial Class price
    Inherits System.Web.UI.Page

    Protected Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles Me.Load
        Dim Q As String = Request.QueryString("p")
        Dim L() As String = File.ReadAllLines(Server.MapPath("Price.txt"), Encoding.Default)
        For i As Integer = 0 To L.Length - 1
            Dim C() As String = L(i).Split(",")
            If C(0) = Q Then
                Response.Write(C(1))
                Response.End()
                Exit For
            End If
        Next
    End Sub
End Class
```

Request 與 Response

最前面匯入 System.IO 是為了要使用 File 物件讀取 Price.txt，File 的 ReadAllLines 蠻好用的，可以直接將分行的文字檔案變成字串陣列。Page_Load 副程式首先用 Request 物件取得使用者提出的查詢字串 QueryString 中稱為『p』的一個變數。所謂的 Request 就是使用者對網頁提出的要求，通常是網址字串後面加入的『?p=xx』這樣格式的問題，以此案例來說，變數 p 代表價錢，xx 就是商品名稱，譬如牛肉麵；所以網頁取得的問題就是：『牛肉麵是多少錢？』

接下來網頁就去讀價目表 Price.txt，此時使用 File 物件的 ReadAllLines 方法，讀取資料行成為字串陣列 L，其中的 Server.MapPath 方法會解析本網站的某個檔案在伺服器電腦中的實際磁碟路徑，因為 VB 程式讀資料時不能看網址，而是必須有實體的磁碟路徑，所以這樣寫是必須的！否則就要自己寫出詳細路徑，如 C:\Server\Price.txt 這樣的檔名，而且如果網站搬家時，你的程式碼也必須跟著修改了！

讀出 L 陣列之後就是一一解析，找到問題中問到的牛肉麵價格，每行價格資料會被 Split 函數以","分成欄位字串陣列 C，C(0)代表商品名稱，C(1)就是該商品的價格。找到正確的 C(0)之後就以 Response.Write(C(1))回應價格給使用者。此地的『回應(Response)』其實就是寫出回覆使用者的網頁！而且用 Response.End 作結束，如果設計的網頁上還有其他的內容，就完全不會繼續傳給使用者。

說『網頁』好像很嚴重，其實就是一個價格字串而已！以此例來說，就只有"80"兩個字。如果這個網頁只負責這件事情，其實可以將原始碼頁面內預設產生的網頁內容(原始碼)全部殺光！

建立提問的視窗程式

接下來請另外建立一個 VB 程式專案，不是網站哦！在表單上建立如下的提問介面：



網址標籤下方的 TextBox 是用來放查詢網頁網址的地方，稍後我們測試時必須將虛擬伺服器產生的，也就是前面製作的 Price.aspx 的網址貼上去，如果網站實際上網運作時，當然就可以把這個網址固定為網頁的實際網址，不必每次重複輸入了！接下來是一個 ComboBox 加入各種商品的名稱，選定商品後應該按下『詢價』按鍵到網站查價錢。下方再加入一個標籤及 TextBox 用來顯示詢問網站獲得的價格。

組織 QueryString 的程式

接著按下『詢價』按鍵寫程式如下：

```
Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) _
    Handles Button1.Click
    '組織網址與查詢字串編碼
    Dim Q As String = TextBox1.Text + "?p=" + Uri.EscapeDataString(ComboBox1.Text)
    '取得暫存檔檔名
    Dim tmp As String = My.Computer.FileSystem.GetTempFileName
    My.Computer.FileSystem.DeleteFile(tmp) '刪除暫存檔
    My.Computer.Network.DownloadFile(Q, tmp) '下載回傳網頁存到暫存檔
    TextBox2.Text = My.Computer.FileSystem.ReadAllText(tmp) '讀取回傳資料
End Sub
```

程式首先是組織出一個網址加上查詢字串 QueryString 需要的字串 Q。TextBox1.Text 是目標網頁的網址，之後加上『?p=』是問題的種類。值得注意的部分是：此例中的 ComboBox1.Text 都是中文字，不能直接加到 QueryString 中！因為網址傳輸使用的 HTTP 通訊協定是看不懂中文的，必須使用特殊的編碼方式，就是 Uri 物件的 EscapeDataString 方法。大概現在網際網路太常用了，使用 Uri 物件甚至不必匯入特別的命名空間。經過這個方法翻譯出來的東西看起來很像一堆亂碼，譬如你可以試試看在 Yahoo 搜尋引擎打入關鍵字搜尋『台灣』，網址列會變成這樣：

```
http://tw.search.yahoo.com/search?p=%E5%8F%B0%E7%81%A3&fr=yfp&ei=utf-8&v=0
```

其中的『%E5%8F%B0%E7%81%A3』其實就是『台灣』的 http 編碼了！如果你直接在網址列打『…?p=台灣』，其實網站是看不懂的！你可以試試看，雖然不會出現錯誤訊息，但跑出來的搜尋結果會完全與『台灣』兩個字無關。有點奇怪的是：回去看看 Price.aspx 的程式碼，好像沒有解譯這種編碼的動作？事實上是解譯回正常字串的動作已經包含在 Request.QueryString 的方法內部了，放心好了，網頁程式看得懂的。

與網站對答的程式

與網站對答的程式是這幾行：

```
Dim tmp As String = My.Computer.FileSystem.GetTempFileName
My.Computer.FileSystem.DeleteFile(tmp) '刪除暫存檔
My.Computer.Network.DownloadFile(Q, tmp) '下載回傳網頁存到暫存檔
TextBox2.Text = My.Computer.FileSystem.ReadAllText(tmp) '讀取回傳資料
```

先用 My 捷徑取得一個暫存檔名 tmp，意思是我們需要一個用過即丟的檔案，還必須先清空內容(刪除它)，再呼叫 My 捷徑的網路(Network)下載檔案(DownloadFile)的功能，將網址傳到網站並承接回傳的網頁到暫存檔 tmp，在此就是詢問商品的價格了！最後再將 tmp 的內容讀出寫到 TextBox2 就完工了！


好像有點小麻煩？原因是經過網際網路 http 通訊回傳的東西不是由瀏覽器直接接收，就是要下載到一個磁碟檔案，我們如果不在專案內使用 WebBrowser 就要存檔，再讀入 VB 專案。還好，仍然順利完成了與網站的對答，程式還算簡短。

測試詢答

接下來請先啟動網頁 `Price.aspx` 的偵錯(F5)，出現一個空白網頁時將它的網址列複製下來。在我的環境測試時網址是這樣的：`http://localhost:49346/Server/price.aspx`。後面的"`Server/price.aspx`"是目錄與網頁的名稱應該沒有問題；前面的『`http://`』是使用的通訊協定也沒問題(網頁都是一樣的)；但中間的『`localhost: 49346`』就需要解釋一下了！

`localhost` 的意義是本機伺服器，如果你的電腦有預設的網站，在瀏覽器網址列打入 `http://localhost` 就會顯示該網站了！『`: 49346`』則是另外指定使用的通訊埠(Port)。一般的網頁伺服器會預設使用通訊埠 80，為了避免與其衝突，Visual Web Developer 為寫作中的網頁建立虛擬伺服器時，會將 Port 設定為系統中未使用的一個 Port 號碼。按照規定：Port 可以使用的範圍是 1~65535，1024 以下區域是作業系統宣稱的保留區，盡量不要用，以免衝突；很多常見的軟體則會集中使用號碼為幾千的 Port，所以自己寫程式時要用到 Port 最好取到更大的幾萬數值區間比較安全。

接下來啟動視窗程式，執行它，將網址貼到程式的 `TextBox`，選擇一項產品，按下詢價就可以看到從網站查到的價格，如下圖：



這個程式只是示範通訊機制之用，實務運作時其實可以將問答程式封裝到一個副程式之中，譬如一個會回傳價格的 `Function`，概略如下：

```
Function Price(URL As String, Pname As String) As Integer
    '組織網址與查詢字串編碼
    Dim Q As String = URL + "?p=" + Uri.EscapeDataString(Pname)
    '取得暫存檔檔名
    Dim tmp As String = My.Computer.FileSystem.GetTempFileName
    My.Computer.FileSystem.DeleteFile(tmp) '刪除暫存檔
    My.Computer.Network.DownloadFile(Q, tmp) '下載回傳網頁存到暫存檔
    Dim P As Integer = Val(My.Computer.FileSystem.ReadAllText(tmp)) '讀取回傳資料
    Return P
End Function
```

之後程式內要查詢商品價格時，只要使用這個 `Function` 傳入網址 `URL` 與商品名稱 `Pname` 即可，很方便的！

專案完整程式碼

ASP.NET 網站

Price.aspx.vb :

```
Imports System.IO
Partial Class price
    Inherits System.Web.UI.Page
    '接受詢價的網頁
    Protected Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles Me.Load
        Dim Q As String = Request.QueryString("p") '接收查詢字串中的產品名稱
        '讀取價目表
        Dim L() As String = File.ReadAllLines(Server.MapPath("Price.txt"), Encoding.Default)
        For i As Integer = 0 To L.Length - 1
            Dim C() As String = L(i).Split(",")
            If C(0) = Q Then '找到符合名稱的產品
                Response.Write(C(1)) '回應價格
                Response.End() '網頁結束
                Exit For '跳出迴圈
            End If
        Next
    End Sub
End Class
```

Windows 專案

Form1.vb :

```
Public Class Form1
    '詢價按鈕
    Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) _
        Handles Button1.Click
        '組織網址與查詢字串編碼
        Dim Q As String = TextBox1.Text + "?p=" + Uri.EscapeDataString(ComboBox1.Text)
        '取得暫存檔檔名
        Dim tmp As String = My.Computer.FileSystem.GetTempFileName
        My.Computer.FileSystem.DeleteFile(tmp) '刪除暫存檔
        My.Computer.Network.DownloadFile(Q, tmp) '下載回傳網頁存到暫存檔
        TextBox2.Text = My.Computer.FileSystem.ReadAllText(tmp) '讀取回傳資料
    End Sub
End Class
```